Lectures on Paper Machine Control

Åström, Karl Johan

1973

Document Version:
Publisher's PDF, also known as Version of record

Link to publication

Citation for published version (APA):
Åström, K. J. (1973). *Lectures on Paper Machine Control*. (Research Reports TFRT-3102). Department of Automatic Control, Lund Institute of Technology (LTH).

Total number of authors:
1

LECTURES ON PAPER MACHINE CONTROL

Simple Paper Machine Models

K.J. Åström

CONTENTS

1. INTRODUCTION.

In this chapter the results of the previous chapters will be summarized to obtain complete models of a paper machine. Models of different complexity are considered.

The first model PM1 is a paper machine where the head-box is assumed to be under perfect control. The head-box dynamics can then be neglected and the major dynamics arises from the mixing and fibre separation that takes place in headbox wire pit, and associated pump and mixing systems.

The next model PM2 is identical with PM1, but the head-box dynamics is now induced.

The presentations follow the same pattern. The physics is first discussed. This leads to a nonlinear model which is then linearized. A FORTRAN program which makes it possible to compute the model parameters from physical data is also induced in each section.

2. PAPER MACHINE NUMBER ONE (PM1).

A very simple model of a paper machine is described in this section. The model is derived under several simplifying assumptions. The headbox dynamics is neglected. The wire is described by a variable retention model. The dryer is characterized by using the water removal rate as an input signal. The model obtained is of second order where the dynamics corresponds to the storage of fibres in the headbox and wire pit. If it is desired to use the pressure in the last drying section as an input instead of the water removal rate another state variable must be added. The model has five inputs and five outputs.

Inputs and Outputs.

A simplified diagram of the paper machine is shown in Fig. 2.1.



Fig. 2.1 - Simplified diagram of a paper machine.

The system has the following control variables (inputs):

u_1 thick stock flow [m^3/s]

u_2 thick stock fibre concentration [kg/s]

u_3 slice opening [m]

u_4 wire speed [m/s]

u_5 water removal rate in dryer section [kg/s]

The following variables are considered as outputs:

y_1 fibre concentration in flow out of headbox [kg/m^3]

y_2 fibre concentration in flow out of wire pit [kg/m^3]

y_3 wet line position measured from the slice [m]

y_4 fibre weight at dry end [kg/m^2]

y_5 water to fibre ratio at dry end

The model is derived from simple mass balances. The major assumptions made to derive the model are given below. For a more detailed discussion of the different subprocesses we refer to the previous chapters.

Headbox Model.

It is assumed that the level and the jet velocity controls are perfect so that the inflow to the headbox always equals the outflow and that the jet velocity always equals the wire speed.

Wet End Model.

It is assumed that the mixing which occurs from the fan pump to the headbox outlet can be characterized as a perfect mixing in a volume V_1 and that the mixing in the wire pit can be characterized as a perfect mixing in a volume V_2. The wire is characterized by a variable retention model. It is assumed that no fibres leave the short circulation, i.e. that the fibre concentration in the wire pit overflow is zero. A fibre balance for the fan pump-headbox system then gives

$$V_1 \frac{dc_1}{dt} = - q_1c_1 + (q_1-q_0)c_2 + q_0c_0 \qquad (2.1)$$

where q_0 is the thick stock flow, q_1 is the flow out of the headbox. The fibre concentrations in thick stock, headbox and wire pit are c_0, c_1 and c_2 respectively. The flow q_0 and the fibre concentration c_0 are control variables. The flow out of the headbox is determined by slice opening and wire speed. Hence

$$q_0 = u_1$$
$$c_0 = u_2$$
$$q_1 = bu_3u_4$$

where b is the machine width.

A fibre balance for the wire pit gives

$$V_2 \frac{dc_2}{dt} = [1 - r(w)]q_1c_1 - (q_1-q_0)c_2 \qquad (2.2)$$

where r is the average retention which is a function of the basis weight a couch w which is given by

$$w = \frac{q_1 c_1 r(w)}{bu_4} = u_3 x_1 r(w) \tag{2.3}$$

Wet Line Position.

Since the model will be used for control studies it is of interest to model the wet line position. This is done as follows. Let ρ denote the density of the fibre mat. The mat thickness at a given distance from the slice is then given by

$$d_m = \frac{w}{\rho} \tag{2.4}$$

Since the fibre mat is obtained by drainage we get

$$w = c_1(u_3 - d_s) \cdot r(w)$$

where d_s is the height of the slurry. Equating d_s and d_m we find that the basis weight at the wet line satisfies the equation

$$w = c_1 r(u_3 - w/\rho) \tag{2.5}$$

Hence

$$w = \frac{c_1 u_3 r(w)}{1 + c_1 r(w)/\rho} \tag{2.6}$$

The wet line position is now implicitely determined from the drainage equation

$$\frac{dw}{dt} = c_1\bar{r}(w)v(w) \tag{2.7}$$

Compare with equation (5.3) of Chapter 2. To obtain an approximative expression for the wet line the retention is put equal to a constant and the drainage velocity is assumed inversely proportional to w in (2.7). The drainage equation (2.7) then reduces to

$$\frac{dw}{dt} = c_1 \frac{k}{w}$$

Integration of this equation gives

$$\frac{1}{2} w^2 = c_1 kt \, ,$$

The drainage time is thus

$$t = \frac{w^2}{2c_1 k}$$

Assume that there is uniform drainage on the wire and note that the wire speed is u_4. The wet line position is then given by

$$y_3 = u_4 t = \frac{w^2 u_4}{2c_1 k} \doteq \frac{c_1 u_3^2 r^2(w)}{2k\left(1 + c_1 r(w)/\rho\right)^2} \tag{2.8}$$

The sheet density at the wet line position is of the order of magnitude of 200 kg/m^3. Considering the very crude approximations made, we then put

$$y_3 \approx \frac{c_1 u_3^2 r^2(w)}{2k}$$

Model for Presses.

To model the presses it is simply assumed that the water to fibre ratio after the presses is constant f_p.

Model for Dryer Section.

A water balance for the dryer section gives

$$f_d \cdot w = f_p \cdot w - u_5$$

where f_p is the water to fibre ratio after the presses and f_d that at the dry end. The input u_5 is the water removal rate in the dryer. If it is preferred to have the pressure in the last dryer section as an input a simple dryer dynamics must be added.

Summary.

Choose the state variables as the fibre concentrations at the headbox and wire pit outlets. Summing up we then find that the paper machine can be described by the equations

$$\frac{dx_1}{dt} = -\frac{bu_3u_4x_1}{V_1} + \frac{(bu_3u_4-u_1)x_2}{V_1} + \frac{u_1u_2}{V_1}$$

$$\frac{dx_2}{dt} = \frac{b[1-r(w)]u_3u_4x_1}{V_2} - \frac{(bu_3u_4-u_1)x_2}{V_2}$$

$$y_1 = x_1$$

$$y_2 = x_2$$

$$y_3 = \frac{u_3^2 x_1 r^2(w)}{2k}$$

$$y_4 = w = u_3 x_1 r(w)$$

$$y_5 = f_d = f_p - \frac{u_5}{b u_3 u_4 x_1 r(w)}$$

Notice that the derivatives are implicitely given due to the implicite equation for the basis weight.

Parameters.

The model of the paper machine is characterized by 5 parameters namely

V_1 effective fan pump-headbox mixing volume [m^3]

V_2 effective wire pit mixing volume [m^3]

b machine width [m]

k parameter in formula for drainage rate [kg $m^{-1}s^{-1}$]

f_p water to fibre ratio after presses

and one function

r(w) average retention

The parameter k and the retention function r depend on the properties of the pulp and will thus vary for different paper grades. The parameters V_1, V_2 and b can be expected to be independent on the paper grade. The parameter f_p will depend both on the operating conditions, in particular on the state of the felts in the presses, and on the properties of the pulp.

To specify the operating conditions it is also necessary to know the values of the five control variables.

Linearization.

The nonlinear model will now be linearized around a steady state operating point. Assuming that the control variables are constant we find from (2.1) and (2.2) that the stationary values of the state variables are given by the linear equation

$$\begin{bmatrix} -q_1 & (q_1-q_0) \\ (1-r)q_1 & -(q_1-q_0) \end{bmatrix} \begin{bmatrix} c_1 \\ c_2 \end{bmatrix} = \begin{bmatrix} q_0 c_0 \\ 0 \end{bmatrix}$$

where

$$q_1 = bu_3u_4$$

$$q_0 = u_1$$

$$c_0 = u_2$$

The linear equation given above can always be solved if $r \neq 0$. The solution is

$$c_1 = \frac{q_0 c_0}{rq_1}$$

$$c_2 = \frac{(1-r)q_1q_0c_0}{rq_1(q_1-q_0)} = \frac{(1-r)q_1}{q_1 - q_0} \cdot c_1$$

The steady state basis weight is given by

$$w = \frac{q_1c_1r}{bv} = \frac{q_0c_0}{bv} = \frac{u_1u_2}{bu_4} .$$

This formula implies that all fibres entering the system as thick stock will have as finished paper or that the fibre flow to the long circulation is neglected.

To carry out the linearization it is necessary to obtain the partial derivatives of the basis weight w with respect to the state variables. We have

$$\frac{\partial w}{\partial x_1} = u_3r + u_3x_1r' \frac{\partial w}{\partial x_1} .$$

Hence

$$\frac{\partial w}{\partial x_1} = \frac{u_3r}{1 - wr'/r}$$

Similarly

$$\frac{\partial w}{\partial u_3} = \frac{x_1r}{1 - wr'/r}$$

The linearization can now be carried out in a straightforward way. The coefficients in the standard form are as follows

$$a_{11} = -\frac{bu_3u_4}{V_1} = -\frac{q_1}{V_1}$$

$$a_{12} = \frac{bu_3u_4 - u_1}{V_1} = \frac{q_1 - u_1}{V_1} = \frac{q_1 - q_0}{V_1}$$

$$a_{21} = \frac{q_1}{V_1}\left[1 - \frac{r}{1 - wr'/r}\right]$$

$$a_{22} = -\frac{bu_3u_4}{V_2} = -\frac{q_1 - u_1}{V_2} = -\frac{q_1 - q_0}{V_2}$$

$$b_{11} = \frac{u_2}{V_1} = \frac{c_0}{V_1}$$

$$b_{12} = \frac{u_1}{V_1} = \frac{q_0}{V_1}$$

$$b_{13} = \frac{b(-x_1+x_2)u_4}{V_1}$$

$$b_{14} = \frac{b(-x_1+x_2)u_3}{V_1}$$

$$b_{21} = \frac{x_2}{V_2}$$

$$b_{23} = -\frac{q_1x_1}{V_2u_3} \cdot \frac{wr'}{1 - wr'/r} - \frac{q_0c_2}{u_3V_2}$$

$$c_{11} = 1$$

$$c_{22} = 1$$

$$c_{31} = \frac{u_3^2 r^2}{2k} \cdot \frac{1 + wr'/r}{1 - wr'/r} = \frac{\ell}{x_1} \cdot \frac{1 + wr'/r}{1 - wr'/r}$$

$$c_{41} = \frac{u_3 r}{1 - wr'/r}$$

$$c_{51} = \frac{u_5}{bu_3 u_4 x_1^2 r} \cdot \frac{1}{1 - wr'/r} = \frac{f_p - f_d}{x_1} \cdot \frac{1}{1 - wr'/r}$$

$$d_{33} = \frac{u_3 x_1 r^2}{k} \cdot \frac{1}{1 - wr'/r} = \frac{2\ell}{u_3} \cdot \frac{1}{1 - wr'/r}$$

$$d_{43} = \frac{x_1 r}{1 - wr'/r}$$

$$d_{53} = \frac{u_5}{bu_3^2 u_4 x_1 r} \cdot \frac{1}{1 - wr'/r} = \frac{f_p - f_d}{u_3} \cdot \frac{1}{1 - wr'/r}$$

$$d_{54} = \frac{u_5}{bu_3 u_4^2 x_1 r} = \frac{f_p - f_d}{u_4}$$

$$d_{55} = -\frac{1}{bu_3 u_4 x_1 r} = -\frac{f_p - f_d}{u_5}$$

Parameters.

The linearized model is completely characterized by 12 parameters, namely 7 physical parameters:

V_1 effective fan pump-headbox mixing volume

V_2 effective wire pit mixing volume

b machine width

k parameter in formula for drainage rate

f_p water to fibre ratio after presses

r average retention

r' derivative of average retention with respect to basis weight,

and steady state values of 5 control variables which determine the operating condition:

u_1 thick stock flow q_0

u_2 thick stock fibre concentration c_0

u_3 slice opening

u_4 wire speed v

u_5 water removal rate in the dryer section.

It is often convenient to give some parameters indirectly. The steady state position ℓ of the wet line can be given instead of the drainage parameter k and the water to fibre ratio at the dry end f_d can be given instead of u_5. A FORTRAN subroutine which evaluates the parameters of the linearized model from the parameters given above is listed below.

```
      SUBROUTINE PM1(A,B,C,D)
C
C     THIS SUBROUTINE GIVES THE ELEMENTS OF THE MATRICES
C     A,B,C AND D FOR A SIMPLE PAPER MACHINE MODEL.
C     THE MODEL HAS TWO STATE VARIABLES.
C
C     AUTHOR K J ASTROM 72/12/03
C
C     X1=C1   FIBRE CONCENTRATION IN FLOW OUT OF HEADBOX  (KG/M3)
C     X2=X2   FIBRE CONCENTRATION IN FLOW OUT OF WIRE PIT (KG/M3)
C
C     FIVE CONTROL VARIABLES
C
C  *  U1=Q0   THICK STOCK FLOW                  (M3/S)
C  *  U2=C0   THICK STOCK FIBRE CONCENTRATION   (KG/M3)
C  *  U3=D1   SLICE OPENING                     (M)
C  *  U4=V    WIRE SPEED                        (M/S)
C     U5      WATER REMOVAL RATE IN DRYER       (KG/S)
C
C     FIVE OUTPUTS
C
C     Y1=X1
C     Y2=X2
C  *  Y3=L    WET LINE POSITION MEASURED FROM SLICE    (M)
C     Y4=W    FIBRE WEIGHT PER UNIT AREA AT DRY END    (KG/M3)
C  *  Y5=FD   WATER TO FIBRE RATIO AT DRY END
C
C     THE MODEL IS CHARACTERIZED BY 12 COEFFICIENTS, NAMELY
C
C     SEVEN PHYSICAL PARAMETERS
C
C  *  V1      EFFECTIVE FANPUMP-HEADBOX MIXIMG VOLUME (M3)
C  *  V2      EFFECTIVE WIRE PIT MIXING VOLUME        (M3)
C  *  B1      MACHINE WIDTH                           (M)
C     K       PARAMETER IN FORMULA FOR DRAINAGE RATE
C  *  FP      WATER TO FIBRE RATIO AFTER PRESSES
C     R       AVERAGE RETENTION FACTOR
C     R1      DERIVATIVE OF AVERAGE RETENTION WITH RESPECT
C             TO BASIS WEIGHT
C
C     AND THE VALUES OF 5 CONTROL VARIABLES ABOVE
C
C     SOME PARAMETERS ARE CONVENIENTLY GIVEN INDIRECTLY
C
C     K       THROUGH L
C     U5      THROUGH FD
C     R AND R1 THROUGH
C  *  ALP     INITIAL RETENTION
C  *  GAM     PARAMETER IN RETENTION FUNCTION
C
      REAL K,L
C
      DIMENSION A(1,1),B(1,1),C(1,1),D(1,1)
C
      COMMON /DAT/   V1,V2,B1,FP,ALP,GAM,Q0,C0,D1,V,L,FD
      COMMON /OPCON/ U(5),Y(5)
      COMMON /PAR/ Q1,R,R1,K
C
C     COMPUTE STATIONARY STATES
C
      Q1=D1*B1*V
      W=Q0*C0/B1/V
      R=GAM*W/ALOG((EXP(GAM*W)-1.+ALP)/ALP)
      R1=R/W*(1.-R*EXP(GAM*W)/(EXP(GAM*W)-1.+ALP))
      C1=Q0*C0/R/Q1
      C2=C1*(1.-R)*Q1/(Q1-Q0)
C
C     COMPUTE INDIRECTLY GIVEN PARAMETERS
C
      K=D1*D1*C1*R*R/2./L
      U(5)=(FP-FD)*B1*D1*V*C1*R
C
C     STATIONARY VALUES OF CONTROL VARIABLES
C
      U(1)=Q0
      U(2)=C0
      U(3)=D1
      U(4)=V
C
C     STATIONARY VALUES OF OUTPUTS
C
      Y(1)=C1
      Y(2)=C2
      Y(3)=L
      Y(4)=W
      Y(5)=FD
C
C     COMPUTE ELEMENTS OF DYNAMICS MATRIX A
C
      A(1,1)=-Q1/V1
      A(1,2)=(Q1-Q0)/V1
      A(2,1)=Q1/V2*(1.-R/(1.-W*R1/R))
      A(2,2)=-(Q1-Q0)/V2
C
C     COMPUTE ELEMENTS OF INPUT MATRIX B
C
      B(1,1)=C0/V1
      B(1,2)=Q0/V1
      B(1,3)=B1*(-C1+C2)*V/V1
      B(1,4)=B1*(-C1+C2)*D1/V1
      B(2,1)=C2/V2
      B(2,3)=-Q1*C1/(V2*D1)*W*R1/(1.-W*R1/R)-Q0*C2/(D1*V2)
C
C     COMPUTE ELEMENTS OF OUTPUT MATRIX
C
      C(1,1)=1.
      C(2,2)=1.
      C(3,1)=L/C1*(1.+W*R1/R)/(1.-W*R1/R)
      C(4,1)=D1*R/(1.-W*R1/R)
      C(5,1)=(FP-FD)/C1/(1.-W*R1/R)
C
C     D-MATRIX
C
      D(3,3)=2.*L/D1/(1.-W*R1/R)
      D(4,3)=C1*R/(1.-W*R1/R)
      D(5,3)=(FP-FD)/D1/(1.-W*R1/R)
      D(5,4)=(FP-FD)/V
      D(5,5)=(FP-FD)/U(5)
C
      RETURN
      END
```

Example.

To get a feeling for the orders of magnitude involved a numerical example will be considered. The following numbers were given for a kraft paper machine:

$V_1 = 10\ m^3$	$q_0 = 0.089\ m^3/s$
$V_2 = 100\ m^3$	$c_0 = 27\ kg/m^3$
$b = 6\ m$	$d = 0.0347\ m$
$\alpha = 0.5$	$v = 6\ m/s$
$\gamma = 50\ m^2/kg$	$\ell = 10\ m$
$f_p = 2$	$f_d = 0.08$

Using the subroutine PM1 we then find the following numerical values of some machine parameters:

$q_1 = 1.25\ m^3/s$	flow out of headbox
$r = 0.83$	average retention
$c_1 = 2.31\ kg/m^3$	fibre concentration in flow out of headbox
$c_2 = 0.42\ kg/m^3$	fibre concentration in wire pit
$w = 0.0668\ kg/m^2$	dry basis weight

The matrices defining the linear model are given by

$$A = \begin{bmatrix} -0.125 & 0.116 \\ 2.22\cdot 10^{-4} & -0.0116 \end{bmatrix}$$

$$B = \begin{bmatrix} 2.7 & 0.0089 & -6.82 & -0.0394 & 0 \\ 0.00419 & 0 & -0.136 & 0 & 0 \end{bmatrix}$$

$$C = \begin{bmatrix} 1 & 0 \\ 0 & 1 \\ 5.89 & 0 \\ 0.0341 & 0 \\ 0.980 & 0 \end{bmatrix}$$

$$D = \begin{bmatrix} 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 680 & 0 & 0 \\ 0 & 0 & 2.27 & 0 & 0 \\ 0 & 0 & 65.3 & 0.320 & 0.416 \end{bmatrix}$$

The stepresponses of this model are shown in Fig 2.2 and Fig 2.3.



Fig 2.2 Normalized stepresponses for the linearized papermachine model. The input amplitudes are chosen equal to the steady state values and the outputs are also normalized with respect to their steady state values.



Fig 2.3 Normalized stepresponses in moisture content for the linearized paper machine model. The input amplitudes are chosen equal to the steady state values and the outputs are normalized with respect to their steady state operating levels.

APPENDIX A - PM1 SUMMARY.



u_1 thick stock flow [m^3/s]

u_2 thick stock fibre concentration [kg/m^3]

u_3 slice opening [m]

u_4 wire speed [m/s]

u_5 water removal rate [kg/s]

$y_1=x_1$ fibre concentration in headbox [kg/m^3]

$y_2=x_2$ fibre concentration in wire pit [kg/m^3]

y_3 wet line position [m]

y_4 fibre weight [kg/m^2]

y_5 water to fibre ratio

$$A = \begin{bmatrix} -0.125 & 0.116 \\ 2.2\cdot 10^{-3} & -0.0116 \end{bmatrix}$$

$$B = \begin{bmatrix} 2.7 & 0.0089 & -6.82 & -0.0394 & 0 \\ 0.00419 & 0 & -0.136 & 0 & 0 \end{bmatrix}$$

$$C = \begin{bmatrix} 1 & 0 \\ 0 & 1 \\ 5.89 & 0 \\ 0.0341 & 0 \\ 0.980 & 0 \end{bmatrix} \qquad D = \begin{bmatrix} 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 680 & 0 & 0 \\ 0 & 0 & 2.27 & 0 & 0 \\ 0 & 0 & 65.3 & 0.320 & 0.416 \end{bmatrix}$$